Hamilton: Der harmonische Oszillator

Zunächst müssen wir mal den Speicher entleeren und nötige Pakete laden. Das braucht Dich nicht zu interessieren...

```
> restart; with
> (plots):
```

```
Warning, the name changecoords has been redefined
```

Und los! Wir wissen daß die Wirkung einer Kurve $\mathrm{x}(t)$ gegeben sein soll durch:

$\mathrm{S}(y) = \int_{t_1}^{t_2} E_{kin}(\frac{d}{dt}\,\mathrm{x}(t)) - E_{pot}(\mathrm{x}(t),\, t)\,dt$

Gib also hier die Funktion F ein!

```
> F:= (u,v,w) -> m/2* v^2-k/2*u^2;
```

$$F := (u,\, v,\, w) \to \frac{1}{2}\, m\, v^2 - \frac{1}{2}\, k\, u^2$$

Jetzt lassen wir Maple die partielle Ableitung nach dem ersten Argument von F berechnen:

```
> D[1](F)(u,v,w);
```

$$-k\, u$$

Beachte, wie man Maple das mitteilt:

```
> D
```

ist der Differentiationsbefehl in Maple.

```
> [1]
```

bedeutet, daß man nach dem ersten Argument partiell ableiteten will,

```
> (F)
```

ist die abzuleitende Funktion und

```
> (u,v,w)
```

ist die Stelle, an der die Ableitung ausgewertet werden soll. Das heißt aber, Maple macht aus einer Funktion wie F durch

```
> D[1](F)
```

wieder eine Funktion mit drei Argumenten!

```
> D[1](F);
```

$$(u,\, v,\, w) \to -k\, u$$

Jetzt brauchen wir das für die Euler-Langrange-Gleichung an der Stelle ($\mathrm{y}(x)$, $\frac{d}{dx}\,\mathrm{y}(x)$, x). Auch wenn Maple die Funktion $\mathrm{y}(x)$ nicht kennt können wir mit ihrer Ableitung arbeiten, indem wir schreiben:

```
> term1:=D[1](F)( x(t), D(x)(t), t);
```

$$\mathit{term1} := -k\, \mathrm{x}(t)$$

Na das geht ja noch; hätte man auch von selbst drauf kommen können! Nun aber zum nächsten Teil, die partielle Ableitung von F nach dem zweiten Argument.

```
> term2:=D[2](F)( x(t), D(x)(t),t);
```

$$\mathit{term2} := m\, \mathrm{D}(x)(t)$$

Jetzt müssen wir das obenstehende noch nach x differenzieren. Das machen wir jetzt mit dem Befehl

```
> diff
```

(weil wir keine Funktion mehr vorliegen haben, sondern nur einen Term, in dem ein x vorkommt):

```
> term3:=diff( term2, t);
```

$$\mathit{term3} := m\, (\mathrm{D}^{(2)})(x)(t)$$

Puh, das sieht ganz schön kompliziert aus. Aber was haben wir denn auch anderes erwartet! Ein Glück, daß wir Maple haben! Dann stellen wir mal die Euler-Lagrange-Gleichung auf! Der

```
> simplify
```

-Befehl hilft uns, das Ganze etwas einfacher darstellen zu lassen!

```
> simplify(term1-term3=0);
```

$$-k\,\mathrm{x}(t) - m\,(\mathrm{D}^{(2)})(x)(t) = 0$$

Jetzt wird es spannend! Vielleicht haben wir Glück, und Maple findet eine Lösung. Wer verwenden den

```
> dsolve
```

-Befehl zum Auflösen der Differentialgleichung nach y(x):

```
> dsolve(term1-term3=0, x(t));
```

$$\mathrm{x}(t) = \_C1\,\sin(\frac{\sqrt{k}\,t}{\sqrt{m}}) + \_C2\,\cos(\frac{\sqrt{k}\,t}{\sqrt{m}})$$

Jetzt zeigt Maple veschiedene Möglichkeiten für die Lösungen an. Welches nun wirklich die richtige Lösung ist, muß man sich anderweitig überlegen. _C1, _C2, .. sind nur Integrationskonstanten. Man muß sie aus den Anfangsbedingungen bestimmen.

Lenkt man die Feder z.B. am Anfang (t=0) um x(0)=1 aus und läßt sie los (Körper in Ruhe: D(x)(0)=0), so bekommt man:

```
> C1:='C1': C2:= 'C2':
> x1:= t->C1*cos(sqrt(k*m)*t/m)+C2*sin(sqrt(k*m)*t/m);
```

Nun kann man irgendwelche Bedingungen einsetzen, z.B. soll y1(3) = 4 und y1(2) = 1 gelten. Diese Gleichungen kann man so schreiben...

```
> glg1:= x1(0)=1;
> glg2:= D(x1)(0)=0;
```

$$x1 := t \to C1\,\cos(\frac{\sqrt{k\,m}\,t}{m}) + C2\,\sin(\frac{\sqrt{k\,m}\,t}{m})$$

$$glg1 := C1 = 1$$

$$glg2 := \frac{C2\,\sqrt{k\,m}}{m} = 0$$

... und Maple löst sie sogar für uns!

```
> C1:=solve(glg1, C1);
> C2:=solve(glg2, C2);
```

$$C1 := 1$$

$$C2 := 0$$

Wir bekommen:

```
> x1(t);
```

$$\cos(\frac{\sqrt{k\,m}\,t}{m})$$